10. gyakorlat

Taskok

Egy taskot úgy kell elképzelni, mintha egy önálló program lenne. Van neki saját kódja és saját memóriaterülete (Adában a taszkok elérhetik a globális változókat is, így használhatnak közös memóriaterületet, FIGYELEM ez egy veszélyes művelet!!). A taskok lényegében párhuzamosan futnak, (azaz végrehajtásuk úgy ütemeződik, hogy azt mi párhuzamosnak érzékeljük) és kommunikálhatnak is egymással. A kommunikáció úgy történik, hogy a hívó fél meghívja egy task belépési pontját, ha a hívott task ezt elfogadja akkor létrejön a kapcsolat, és egy ideig közösen futnak (azaz közös a kódjuk és ez a kód a hívott félnél van megírva). Két taszk kommunikációját randevúnak hívják. Egy task nem önálló fordítási egység, lehet őket több fileba tenni, de mi az egyszerűség kedvéért minden taszkot a főprogram deklrációs részében definiáljunk.

Task létrehozása - specifikációs rész

task <Tasknév> is

   entry < Belépési_Pont_Név > (

< Paraméterek >); 

   ...

   end <Tasknév>;

Ha egy tasknak nincs belépési pontja akkor elég az alábbi specifikáció:

   task <Tasknév>; 

Task létrehozása - törzsrész

task body <Tasknév> is

   [Deklarációs Rész]

   begin

   [Utasítások]

   end <Tasknév>; 


Ha a törzsrészt külön fileba akarjuk tenni, akkor ide a köv. dolgot kell írni:

TASK BODY <tasknév> IS SEPARATE; 

 
FONTOS!! A specifikációs résznek mindig meg kell előzni a törzsrészt, és minden más olyan task törzsrészét is ahol hivatkozunk rá.



Task típusok

Előfordulhat, hogy egy taszkból több példányt is szeretnénk létrehozni, (pl. közlekedés szimulációban minden gyalogosnak ugyanaz a kódja) ekkor használjuk a task típust. Így egyszer definiálunk egy típus működését, majd tetszőleges számú példányt tudunk belőle létrehozni. A taszk típust diszkriminánsal is elláthatjuk. A specifikációs rész szintakszisa a következő:

task type <Tasktípusnév>[(<Diszkriminánsnév_1> : <Diszktét_Típus_1>[; ...

 [;<Diszkriminánsnév_N> : <Diszktét_Típus_N]])] is

   entry < Belépési_Pont_Név > (< Paraméterek >); 

   ...

   end <Tasknév>; 

 
Természetesen itt sem kötelező belépési pontokat megadni. Ekkor a kód a következő:

TASK TYPE <tasktípusnév>[(<diszkriminánsnév_1> : <diszkrét_típus_1>[; ... [;<diszkriminánsnév_n> : <diszkrét_típus_n]])]; 

 
A törzsrész szintakszisa megegyezik a sima taszkéval

Ezek azonban még csak típusok, ezekből létre kell hozni a konkrét taszkokat.
Ennek két módja van, egyik a deklarációs részben hozzuk létre

<Konkrét_Task> : <Tasktípusnév>; 


Vagy akár tömböt is csinálhatunk belőle, így annyi taszkunk lesz, ahány elemű a tömb

<konkrét_tasktömb> : ARRAY(1..n) OF <tasktípusnév>; 


A másik mód, hogy létrehozunk tasktípusra mutató mutatót, majd a főprogramban vagy valamelyik taskban dinamikusan hozunk létre új taskot:

TYPE <taskmutatótípus> IS ACCESS <tasktípusnév>; 

<taskmutató> : <taskmutatótípus>; 

BEGIN 

... 

<taskmutató> := NEW <tasktípusnév>; 

Taskok elindulása és terminálása

Azok a taskok amelyek önállóak (azaz nem egy task típusból hoztuk őket létre), illetve azok a task típusból létrehozott taskok amelyeket a deklarációs részben hoztuk létre akkor indulnak el, amikor a vezérlés elérkezik az őket tartalmazó deklarációs rész BEGIN-jéhez. Amelyeket dinamikusan hoztuk létre, azok a new utasitás kiadásakor indulnak.
Egy taszk kompletté válik (azaz még létezik, de már nem fut) amikor eljut az utasításai végére, majd terminál, ha minden tőle függésben levő taszk kompletté vált. Kompletté válhat ha terminate utasításhoz ér, vagy kívülről aborttal leállították.



Task attribútumok

	PRIVATE
tasknév'callable
	hamis, ha a task komplett

	tasknév'terminated
	igaz, ha a task terminált

	belépési_pont'Count
	visszaadja, hogy hány taszk várakozik ennél a belépési pontnál. Ezzel csak a task saját belépési pontjait kérdezheti le.


Belépési pontok / hívás fogadás

A belépési pontoknál tud csatlakozni a hívó taszk a fogadóhoz. A belépési pontot a taszk specifikációs részében is fel kell tüntetni:

ENTRY <belépési_pont_név> [(<paraméter_1> : [IN | OUT | IN OUT] <típus_1>[; ... [<paraméter_n> : [IN | OUT | IN OUT] <típus_n>] ])]; 

 A paraméter módjai:

	PRIVATE
IN
	Adatot vár a hívótól

	OUT
	Adatot szolgáltat a hívónak

	IN OUT
	Adatot vár és szolgáltat


Hívás elfogadása / belépési pontok a törzsrészban

1. A hívott fél vezérlése addig várakozik egy belépési pontnál, amíg azt valaki meg nem hívja. Ilyenkor a vezérlés akkor sem ugrik át más belépési ponthoz, ha ott már várakoznak.


accept < Belépési_Pont_Név > (< Paraméterlista >) do 

   <Utasítások>

  end <Belépési_Pont_Név>; 

2. Több belépési pont közül arra ugrik a vezérlés, ahol hívás van. Ha egyszerre több belépési ponthoz érkezik hívás, akkor nem determinisztikusan választ egyet közülük.


select

   accept < Belépési_Pont_Név_1 > (< Paraméterlista >) do 

      <Utasítások>

      end <Belépési_Pont_Név_1>;

or

   accept < Belépési_Pont_Név_2 > (< Paraméterlista >) do 

      <Utasítások>

      end <Belépési_Pont_Név_2>;

   .

      .

      .

or

   accept < Belépési_Pont_Név_N > (< Paraméterlista >) do 

      <Utasítások>

      end <Belépési_Pont_Név_N>;

   [or

   terminate;

   ]

end select; 

 
A TERMINATE ág akkor hajtódik végre, ha már az összes olyan taszk kompletté vált, amelyik a SELECT-ben szereplő belépési pontok valamelyikét hívhatná. Ez arra jó, hogy ha van egy kiszolgáló taszkunk, akkor ezt a SELECT blokkot egy végtelen ciklusba tehetjük, és ha az összes  ide vonatkozó taszk kompletté válik, akkor a kiszolgáló taszk eljut a TERMINATE utasításhoz és ő is terminál. Így egyszerűen helyettesíteni tudunk, egy bonyolult kilépési feltételt.

Lehetőségünk van őrfeltételeket is rendelni a belépési pontokhoz. Ez azt jelenti, hogy a taszk mielőtt elfogadná egy belépési ponthoz értkező hívást, megnézi, hogy az őrfeltétele igaz-e. Ha igaz, elfogadja a hívást, ha hamis, akkor pedig úgy tekinti, hogy ez az ág nem szerepel a SELECT-ben. Őrfeltételt csak a SELECT struktúrában tudunk alkalmazni. (Könnyen meggondolható, hogy az előző esetben nincs értelme az őrfeltételnek (az igaz őrfeltételnek nincs hatása, a hamis pedig �lefagyasztaná" a taszkot).)

select

   [when <Feltétel_1>]   accept < Belépési_Pont_Név_1 > (< Paraméterlista >) do 

      <Utasítások>

      end <Belépési_Pont_Név_1>;

   or

      [when <Feltétel_2>]   accept < Belépési_Pont_Név_2 > (< Paraméterlista >) do 

      <Utasítások>

      end <Belépési_Pont_Név_2>;

   .

      .

      .

      or

      [when <Feltétel_N>]   accept < Belépési_Pont_Név_N > (< Paraméterlista >) do 

      <Utasítások>

      end <Belépési_Pont_Név_N>;

   [or

   terminate;

   ]

 end select; 

  Ha mégis rendszeres feltételvizsgálatra van szükségünk:

LOOP
    SELECT
        when feltétel =>  accept < Belépési_Pont_Név_1 >;
    OR / ELSE 
        delay t;
    END SELECT;
END LOOP; 
A delay előtt else van or helyett, akkor a várakozás alatt nem fogadja az esetleges hívást, azaz a két fél "elkerülheti" egymást, ha a másik fél sem túl türelmes.
Minél kisebb t, annál gyakoribb a vizsgálat, de annál jobban leterheli a processzort. Akár 0.0 is lehet, az már majdnem teljesen folyamatos, ha még az is kevés lenne, akkor a delay t; helyére kerülhet null; utasítás is, ami még sűrűbb feltételvizsgálatot eredményez, de ekkor mindenképp else kell elé. 
Az or delay, or terminate ágak legfeljebb egyszer szerepelhetnek egy selecten belül, ugyanígy csak egy else ág lehet benne, minden más tetszőleges mennyiségben kombinálható.
3. Ha nem akarunk várakozni egy belépési pontnál, akkor azt a következő módon tehetjük meg:


select

   accept < Belépési_Pont_Név > (< Paraméterlista >) do 

      <Utasítások>

      end <Belépési_Pont_Név>;

else

   <Utasítások>

   end select; 

 Tehát itt először megnézzük, hogy vár-e taszk a belépési pontnál, ha igen akkor elfogadjuk a hívását, ha nem vár ott senki akkor pedig az ELSE ágban leírt programrész fog lefutni.



Hívás

A hívásnak is három módja van:

1. Addig vár a hívó, amíg hívását el nem fogadják:

<Tasknév>.<Belépési_Pont_Név>[(<Paraméterek>)]; 

2. Megnézi, hogy rögtön elfogadják-e a hívását, ha igen létrejön a kapcsolat, ha nem akkor mást csinál

select

   <Tasknév>.<Belépési_Pont_Név>[(<Paraméterek>)];

else

   <Utasítások>

   end select; 

3. Megnézi, hogy egy adott időn belül elfogadják-e a hívását, ha igen létrejön a kapcsolat, ha nem akkor mást csinál

select

   <Tasknév>.<Belépési_Pont_Név>[(<Paraméterek>)];

or

   delay <Türelmi_Idő_Másodpercben>;

   <Utasítások>

   end select; 


A DELAY DURATION típusú értéket vár. (A 1.0 átkonvertálódik duration típussá, de ha a f float típusú, akkor a delay f nem megengedett.)
A DELAY utasítást önmagában is ki lehet adni, akkor azt jelenti, hogy a vezérlés a paraméterben kapott másodpercig vár. 



Példa 1. semmi_t.adb -- minimális taszkos program
Írjunk egy olyan taszkot a null utasítás használatával., ami nem csinál semmit.

procedure Semmi_T is


task T;


task body T is 


begin 


  null; 


end;

begin 

  null; 

end;

Példa 2.  ab.adb -- taszk objektum
Írjunk két párhuzamosan futó taszkot, amelyek egyenként 1-től a Positive értéktartomány utolsó eleméig kiírják a pozitív számokat, illetve magukat is azonosítják a kiírt szám mellett.

with text_io; use text_io;

procedure AB is


task A;


task B;


task body A is


begin


  for i in 1..POSITIVE'LAST loop


    Put_Line("A:"&POSITIVE'IMAGE(i));


  end loop;


end A;


task body B is


begin


  for i in 1..POSITIVE'LAST loop


    Put_Line("B:"&POSITIVE'IMAGE(i));


  end loop;


end B;

begin

  null;

end AB;

Példa 3. jj.adb -- taszk típus, entry
Hozzunk létre egy Kiir nevű taszk típust egy stringgel paraméterezhető belépési ponttal. A Kiir tipusú taszkok a paraméterben kapott stringről készítsenek egy stringre mutató pointerbe egy másolatot és 1-től a Positive értéktartomány végéig írják ki a kapott sztringet magával a pozitív számmal együtt.  Futassunk két taszkot párhuzamosan, melyek a "jancsi" ill. "juliska" sztringgel hívják meg a Kiir taszkot. 

with text_io; use text_io;

procedure JJ is


task type Kiiro is --taszk tipus letrehozasa



entry Init ( s: in STRING := "" ); -- parameteres, -- belépési pont


end Kiiro;                               


a, b: Kiiro; --ket taszk deklaráció


task body Kiiro is



type PStr is access STRING;



nev: PStr;


begin


  accept Init ( s: in STRING := "" ) do -- hivas fogadas


  
nev := new STRING(1..s'length);



nev.all := s;


  end Init;


  for i in POSITIVE'range loop


    Put_Line(nev.all & POSITIVE'IMAGE(i));


  end loop;


end Kiiro;

begin


a.Init("Jancsi"); b.Init("Julcsa"); -- ket taszk hivasa kulonbozo parameterekkel

end JJ;

Példa 4. tp.adb -- taszk típus diszkriminánsa, delay

Definiáljunk egy diszkriminánsos Kiir nevű taszk típust, melynek diszkriminánsa egy integer típusú érték (alapérték: 42) legyen a taszk azonosítására.

A Kiir típusú taszkok végtelen ciklus keretében írják ki a saját azonosítójukat. Parancs sori argumentumként megadott idő tartamra adjunk lehetőséget a taszkok futtatásának felfüggesztésére.

with TEXT_IO,ada.integer_text_io,ada.command_line;

 use TEXT_IO,ada.integer_text_io,ada.command_line;

procedure tp is

task type kiir(nev:INTEGER:=42);

task body kiir is

 begin

  loop

   put(nev);

   new_line;

   if argument_count>0 then

    delay duration'value(argument(1));

   end if;

  end loop;

 end kiir;

a:kiir;

b:kiir(1);

c:kiir(2);

begin

 put(0);

 new_line;

end;

